
SAS syntax and output for Chapter 10: Structural Equation Modeling II: Latent Variable Models
Much of the material in this document builds on the SAS sytax and output from earlier chapters. If necessary, refer back to those documents for reminders.

Begin by importing the 'adswan.txt' file for the ADHD symptom example:
	data ch10adhd;
 infile 'c:\adswan.txt' firstobs=2 dlm=' ';
 input id $ swanp1 - swanp18;
 drop id;
run;



As always, one should graph the data and inspect descriptive stats before fitting models.
To reduce clutter, Figure 10.1 and Table 10.1 present a scatterplot matrix and correlations among items 1, 6, 9, 11, and 18 rather than all 18 items.
Thus, a new dataset containing only these 5 variables can be created with
	data temp;
 set ch10adhd;
 keep swanp1 swanp6 swanp9 swanp11 swanp18;
run;


And then the scatterplot and correlation matrix are created using familiar commands from previous chapters.
	proc corr data=temp noprob;
 var swanp1 swanp6 swanp9 swanp11 swanp18;
run;



proc sgscatter data=temp;
 matrix swanp1 swanp6 swanp9 swanp11 swanp18 / diagonal=(histogram);
run;

Of course, one should examine the data for ALL variables, not just this subset of 5 items. The code and ouput below shows these comprehensive results.

	proc corr data=ch10adhd noprob outp=temp;
 var swanp1 - swanp18;
run;



proc sgscatter data=ch10adhd;
 matrix swanp1 - swanp18 / diagonal=(histogram);
run;

As one can see above, a scatterplot matrix for all 18 variables creates a such a large grid of plots that each individual plot is very difficult to decipher.

CFA: model specification
The latent variable models of this chapter can be estimated using the 'calis' procedure, which was introduced in the SAS document for Chapter 9.
The two-factor CFA model in Figure 10.2 can be specified with the following text string.
The "factor" option is used to specify which observed variables (named on the right-hand side of the operator) are indicators of the latent variable named on the left-hand side of the operator:
	proc calis data=ch10adhd;
 factor
 inat ---> swanp1 - swanp9,
 hypter   ---> swanp10 - swanp18;
run;


By default, proc calis will interpret this model specification such that the latent variables are freely covaried and while all error variances are fixed to zero.

CFA: Model estimation
Below, the "method='FIML'" option is used to implement the so-called "full-information maximum likelihood" estimation to account for the presence of incomplete cases. FIML estimation, also known as "direct maximum likelihood," is described in Chapter 11. Unfortunately, proc calis does not support robust maximum likelihood procedure described in Chapter 10 together with FIML.
Finally, the latent variables under the argument ‘pvar’ are included so that the latent variables are scaled by fixing their variances equal to 1 (thus over-riding the default marker-indicator approach):
	proc calis data=ch10adhd method=FIML;
 factor
 inat ---> swanp1 - swanp9,
 hyper   ---> swanp10 - swanp18;
 pvar
 inat = 1,
 hyper = 1;
run;

Overall, this two-factor model does not fit the data well.
The output above also includes extensive results pertaining to the parameter estimates, but because the model fit is not adequate, we should avoid interpreting these estimates.

One-factor model
The one-factor model described in Chapter 10 (Figure 10.5) is specified and estimated below:
	proc calis data=ch10adhd method=FIML;
 factor
 adhd ---> swanp1 - swanp18;
 pvar
 adhd = 1;
run;

As reported in Chapter 10, this model fits the data poorly.

Three-factor model
The three-factor model described in Chapter 10 (Figure 10.6) is specified and estimated below:
	proc calis data=ch10adhd method=FIML;
 factor
 inat ---> swanp1 - swanp9,
 hyper ---> swanp10 - swanp15,
 impul ---> swanp16 - swanp18;
 pvar
 inat = 1,
 hyper = 1,
 impul = 1;
run;

As reported in Chapter 10, this model fits the data poorly.

Bifactor model
The bifactor model described in Chapter 10 (Figure 10.7) is specified below:
	proc calis data=ch10adhd method=FIML;
 factor
 general ---> swanp1 - swanp18,
 spec1 ---> swanp1 - swanp9,
 spec2 ---> swanp10 - swanp18;
 pvar
 general = 1,
 spec1 = 1,
 spec2 = 1;
 cov
 general spec1 spec2 = 3 * 0;
run;


Above, 'spec1' and 'spec2' are the names given to the specific inattention and specific hyperactivity factors, respectively.
The 'cov' option is used to refer to the covariances among the latent variables. By default, these covariances are free parameters. But using '0*' fixes these covariances to 0. The number 3 refers to the 3 covariances.

In the output above, first notice that the bifactor model fits the data well (although proc calis does not provide the robust model fit stats reported in Chapter 10).
Next, under 'Factor Loading Matrix', the 'Estimate' values listed in the left-hand column correspond to the unstandardized factor-loading estimates given in Table 10.2, along with their associated standard errors and t tests. 
Under 'Factor Covariance Matrix', notice that the estimated covariances among the latent variables all equal 0 because they were fixed to 0 in accordance with the a priori specification of the bifactor model.
The 'Intercepts' values are usually ignored in CFA applications.
Finally, remember that the estimates for the observed variables listed under 'Error Variances' correspond to residual variances, not the actual observed sample variances of these variables.

Structural regression models
Import the 'health.dat' file for the healthcare utilization example. This data file is not identical to the one used for the healthcare utilization example presented in Chapter 10; instead, these data were simulated from a multivariate normal population distribution consistent with the correlations, means, and standard deviations given in Table 10.3.
	data ch10dat3;
 infile 'c:\health.dat' firstobs=2 dlm='';
 input id age strss esteem attmar control phyheal menheal druguse timedrs;
 drop id;
run;


The code below creates a scatterplot matrix similar to Figure 10.11:
	proc sgscatter data=ch10dat3;
 matrix age strss esteem attmar control phyheal menheal druguse timedrs / diagonal=(histogram);
run;

Because the current data were simulated from a multivariate normal distribution, the scatterplots above do not show the same potentially non-linear trends and non-normality that were observed in the original data and presented in Figure 10.11.
While we're at it, we should get univariate descriptive stats and take a look at the correlations among the variables:
	proc means data=ch10dat3 maxdec=2 n mean std median min max range skew kurt;
run;



proc corr data=ch10dat3 noprob;
 var age strss esteem attmar control phyheal menheal druguse timedrs;
run;





The results above should be consistent (but not quite identical) with those reported in Table 10.3.

Structural regression models: Measurement model specifiaction and estimation
Following the so-called "two-step" procedure, the model is first specified and estimated as a CFA model:
	proc calis data=ch10dat3 method=MLSB;
 factor
 self ---> esteem attmar control,
 ill ---> phyheal menheal,
 util ---> druguse timedrs,
 agel ---> age = 1,
 stressl ---> strss = 1;
 pvar
 age = 0,
 strss = 0;
run;


The model specification above is much like the CFA model specifications presented earlier.
But as explained in Chapter 10 and shown in Figure 10.13, in this model we need to create single-indicator latent variables for the 'age' and 'stress' observed variables.
Thus, 'agel' is a latent variable measured by 'age' with its factor loading fixed to equal 1 and likewise 'stressl' is a latent variable measured by 'stress' with its factor loading fixed to equal 1.
Finally, the last two lines of the model specification text string fix the error variances of the 'age' and 'stress' observed variables to equal 0. These lines are necessary to define 'age' and 'stress' as perfect indicators of the 'agel' and 'stressl' latent variables.
Next, the measurement model is estimated using the 'method="MLSB"' option to invoke Satorra-Bentler robust model fit stats and standard errors, as explained in Chapter 10. Here, the data are known to be multivariate normal (because they were simulated), but the MLM option is still shown because it should be used with real data.
Fitting this model to the data produces a warning from SAS because of an improper solution, as reported in Chapter 10.
We can look at the results summary not to interpret the estimates but instead to try to figure out the source of the improper solution (the summary below doesn't bother to ask for the optional model fit stats because we already know that the model is problematic):

[bookmark: _GoBack]In the output above, we see that the correlation between the 'ill' and 'util' latent variables is suspiciously strong (completely standardized covariance = correlation = .97).
As reported in Chapter 10, this association is likely due to some redundancy between the 'timedrs' and 'phyheal' observed variables.

Fit Summary

Modeling Info | Number of Complete Observations 49
Number of Incomplete Observations 7

Saturated Model Estimation FIML

NOTE: Saturated mean structure parameters are.

excluded from the computations of fit i

Fit Summary

Modeling Info | Number of Complete Observations 49

Saturated Model -2 Log-Likelihood | 7942 3524
Baseline Model Estimation Converged
Baseline Model Function Value 57.0399
Baseline Model -2 Log-Likelihood | 9468 6198
Baseline Model Chi-Square 15262674
Baseline Model Chi-Square DF 153
Pr > Baseline Model Chi-Square <0001

Absolute Index | Fit Function 489165
-2 Log-Likelihood 81201341
Chi-Square 1777818
Chi-Square DF "7
Pr> Chi-Square 0.0002
ZTest of Wilson & Hilferty 34775
Hoelter Critical N 133
Root Mean Square Residual (RMR) 00708
Standardized RMR (SRMR) 0.0499
Goodness of Fit Index (GFI) 08930

Parsimony Index | Adjusted GFI (AGFI) 08437
Parsimonious GFI 0.
RMSEA Estimate 00559
RMSEA Lower 90% Confidence 00385
RMSEA Upper 90% Confidence 00720
Probability of Close Fit 0.2687
Akaike Information Criterion 82281341
Bozdogan CAIC 8450.1815
Schwarz Bayesian Criterion 83961815
McDonald Centrality 08327

Incremental Index | Bentler Comparative Fit Index 09557
Bentler-Bonett NFI 01424
Bentler-Bonett Non-normed Index 09421
Bollen Normed Index Rho1 08477
Bollen Non-normed Index Delta2 0.9569
James et al. Parsimonious NFI 06756

NOTE: Saturated mean structure parameters are.

Fit Summary

ModelingInfo | Number of Observations 445
Number of Variables 9
Number of Moments 45
Number of Parameters 29
Number of Active Constraints 0
Baseline Model Function Value 1.7662
Baseline Model Chi-Square 7841959
Baseline Model Chi-Square DF 36
Pr > Baseline Model Chi-Square <0001
Absolute Index | Fit Function 0.2060
Chi-Square 91.4477
Chi-Square DF 16
Pr>Chi-Square <0001

Root Mean Square Residual (RMR)  1.2996
Standardized RMR (SRMR) 0.0572
Goodness of Fit Index (GFI) 09574
Parsimony Index | Adjusted GFI (AGFI) 0.8803
Parsimonious GFI 0.4255
RMSEA Estimate 0.1031
RMSEA Lower 90% Confidence Limit| 00831
RMSEA Upper 90% Confidence Limit| 01241
Probability of Close Fit <0001
ECVI Estimate 0.3396
ECVI Lower 90% Confidence Limit 02797
ECVI Upper 90% Confidence Limit 0.4168
Akaike Information Criterion 149.4477
Bozdogan CAIC 297.2919
Schwarz Bayesian Criterion 268.2919
McDonald Centrality 09187
Incremental Index | Bentler Comparative Fit Index 0.8992
Bentler-Bonett NFI 08834
Bentler-Bonett Non-normed Index 07731
Bollen Normed Index Rho1 07376
Bollen Non-normed Index Delta2 09018
James et al. Parsimonious NFI 0.3926

